
Pathex
    

Speed or composability? Choose both!

  
    
  
  What is Pathex?


Pathex is a library for performing fast actions with nested data structures in Elixir.
With pathex you can trivially set, get and update values in structures in a functional manner.
It provides all necessary logic to manipulate data structures in different ways using flexible functional lens pattern.

  
    
  
  Why another library?


Existing methods of accesssing data in nested structures are either slow (like Focus) or do not provide enough functionality (like Access). For example setting the value in structure with Pathex is 70-160x faster than Focus or 2-3x faster than put_in and get_in
You can checkout benchmarks at https://github.com/hissssst/pathex_bench


  
    
  
  Usage


Pathex is really simple and straightforward to use (almost like Enum). You don't need to learn any specific language, just create paths with path and use verbs with them.

  
    
  
  Add it to your module


# This will import path macro and operators and require Pathex
use Pathex
Or just import Pathex


  
    
  
  Create the path


path_to_streets = path :user / :private / :addresses / 0 / :street
path_in_json = path "users" / 1 / "street", :json
This creates closure which can get, set, update and delete values in this path

  
    
  
  Use the path


{:ok, "6th avenue" = street} =
  %{
    user: %{
      id: 1,
      name: "hissssst",
      private: %{
        phone: "123-456-789",
        addresses: [
           [city: "City", street: "6th avenue", mail_index: 123456]
        ]
      }
    }
  }
  |> Pathex.view(path_to_streets)

%{
  "users" => %{
    1 => %{"street" => "6th avenue"}
  }
} = Pathex.force_set!(%{}, path_in_json, street)

  
    
  
  Features


Pathex has a lot of different features and can even compete with code written by hand in terms of efficiency.
Pathex significantly reduces the time to write a code which manipulates nested structure, while
providing efficiency and composability. No more functions like get_users, set_users, update_users! No more XPaths, JSONPaths, CSS Selectors!

  
    
  
  Easy to use


It's not harder to use than Map or Enum! Check out the cheatsheet for common tasks.
Pathex also provides more information about errors than any other tool.
iex(1)> field = :email
iex(2)> Pathex.view!(%{}, path(:users) ~> all() ~> path(:personal / field))
** (Pathex.Error)
  Couldn't find element

    Path:      path(:users) ~> all() ~> path(:personal / :email)

    Structure: %{}

  
    
  
  Fast


Paths are just a set of pattern-matching cases.
This is done to extract maximum efficiency from BEAM's pattern-matching compiler.
# Code for viewing variables for path
path(1 / "y", :map)

# Almost equals to
case input do
  %{1 => %{"y" => res}} ->
    {:ok, res}

  _ ->
    :error
end

  
    
  
  Reusable


One path can be used to update, get, set, delete or update a value in the structure!
And these paths can even be composed together.
This composition is very efficient, there's no need to concatenate lists like Access does.
# User structure
user = %User{
  personal: %{fname: "Kabs", sname: "Rocks"},
  phone: "123-456-789"
}

# Path to username in user structure
username = path(:personal / :fname)

# Get a username
{:ok, "Kabs"} = Pathex.view(user, username)

# Set a username
another_user =
  %User{
    personal: %{fname: "Blabs", sname: "Rocks"},
    phone: "123-456-789"
  } = Pathex.set!(user, username, "Blabs")

# Get all usernames!
import Pathex.Lenses
["Kabs", "Blabs"] =
  [
    user,
    another_user
  ]
  |> Pathex.view!(all() ~> username)
Pathex can be used to manipulate different nested data structures. From GenServer state to HTML or Elixir's AST!

  
    
  
  Extensible


Pathex is built around simple primitive called path-closure, which is a simple closure with clearly defined specification. Anything complying with Pathex.t() spec can be used within Pathex.

  
    
  
  Installation


def deps do
  [
    {:pathex, "~> 2.0"}
  ]
end
Pathex is significantly faster with OTP 26


  
    
  
  Contributions


Welcome! If you want to get your hands dirty, you can check existing TODO's.
By the way
If you have any suggestions or want to change something in this library don't hesitate to open an issue. If you have any whitepapers about functional lenses, you can add them in a PR to the bottom of this readme




  

    
Changelog
    


  
    
  
  2.5.0


Breaking
None
Non-breaking
	Pathex.Accessibility.from_struct/2 to create force_over-friendly lenses for Elixir structures
	Pathex.Accessibility.from_record/3 to create force_over-friendly lenses for records


  
    
  
  2.4.3


	Documentation improvements, typos etc.
	Documentation for VSCode reformatting


  
    
  
  2.4.2


	Fixed :delete op bug for all lens #21
	Improved performance of all lens
	Added Code of Conduct


  
    
  
  2.4.1


	Fixed :delete op bug for all lens #20


  
    
  
  2.4


Negative indexes, lens fixes, force_update nil filling and refactoring
Breaking
	matching force_update fix. Now it doesn't raise an error when called
	filtering force_update fix. Now it doesn't raise an error when called
	Tuple force_update fix. Now it works with indexes and for force_set {}, path(2), 1 it will create {nil, nil, 1}.
	List force_update fix. Now it works with indexes and for force_set [], path(2), 1 it will create [nil, nil, 1].

Non-breaking
	Tuple negative index is now supported for all operations
	List negative indexes now work the same way as positive indexes (but backwards), except -1 which still prepends to the list.


  
    
  
  2.3


Breaking
	Pathex now uses Elixir 1.13 (or higher) version

Non-breaking
	Pathex.t spec fix
	Pathex.Short for shorter path definition
	Pathex.pattern for creating patterns from inlined paths
	Pathex now can be use-d inside functions or anything like this
	Paths inlining is now detected for aliased, imported and macro calls
	Internal spec fixes, dead code eliminations and formatting


  
    
  
  2.2


Breaking
	Changed behaviour of path-closures created with path when working with improper Keyword.t()

Non-breaking
	Improved documentation
	Removed dead code
	Added Pathex.Accessibility module for creating paths in runtime


  
    
  
  2.1


Breaking
	Updating and viewing keywords with star and some lenses now doesn't raise when they're used against non-proper keyword

Non-breaking
	Concatenated paths now can force_over for not only maps
	Fixed debug lens
	Added ability to pass calls and arbitary structures into path macro
	Unrolled some clauses for star and some for extra efficiency


  
    
  
  2.0


Breaking
	Reworked star lens. Now it is less optimistic and returns :error when no values were viewed/updated
	Removed sigils
	Removed deprecated lens id
	Removed deprecated lens either
	Removed recur function

Non-breaking
	compose function for recursive lens
	delete method for all paths, lenses and higher order functions
	inspect method for all paths, lenses and higher order functions
	Matchable updater for lists and maps
	Builders are selected for combination (not for mod as they used to)
	Reworked documentation
	Annotated paths


  
    
  
  1.3.0


Breaking
None! (See deprecated in Non-breaking)
Non-breaking
	Deprecated id lens

	Deprecated either lens

	Fixed bug with concatenation context overlapping

	some lens

	star lens

	matching lens

	filtering lens

	Removed some dead code

	Moved lenses code to separate modules



  
    
  
  1.2.0


Breaking
None!
Non-breaking
	star lens
	all lens


  
    
  
  1.1.0


Breaking
None!
Non-breaking
	||| operator
	stack-optimized version of ~> operator
	stack-optimized version of ||| operator


  
    
  
  1.0.0


Breaking
	force_set/get/set clause in closure was renamed to force_update/view/update
and added a special argument with default value in it

Non-breaking
	alongside macro
	stack-optimized version of &&& operator
	path code generation size assertion
	better documentation format
	id lens
	either lens
	any lens

Yeah, it starts from 1.0.0
I can describe previous versions if anybody needs this. Just open an issue! :)




  

    
Basics
    

This guide will show the basics of Pathex. This is the best place to start learning Pathex.
It won't take more than 5 minutes to read.
As README says, Pathex is a library for working with collections and nested key-value structures.
Pathex uses a powerful functional abstraction called a lens (in Pathex we call it path or path-closure, you'll later see why).
Path is a basically a path to a value in a nested structure, almost like filesystem path.
With this path you can use some verbs to get, set, delete or update the value in the structure.

  
    
  
  Create paths


First of all, we need to know how to create paths. A simple path can be created with the Pathex.path/2 macro.

  
    
  
  For example:


p = path :x / :y
# Note the "/". That's why the library is called Pathex
This path specifies a way to get (or set, or delete or whatever) value 1 from %{x: %{y: 1}}
Note:
You don't have to pin variables used in paths
x = 1
path :some_atom / x  # Right
path :some_atom / ^x # Wrong

Each type of value used in a path corresponds to one or more types of Elixir collections
	Strings, tuples, maps, functions, lists are treated only as Map keys.
	Atoms are treated as Map or Keyword keys.
	Positive integers are treated as Map keys or List and Tuple indexes
	Negative integers are treated as Map keys or first element in List's view or
prepending element in List's update or force_set operations
	Variables are treated the same way as the type they have in runtime. For example, variables containing negative
integers are treated as negative integer values. Unfortunately, it is not possible to know variable type at compile-time,
that's why Pathex generates code for all possible types. See performance tips for more information.


  
    
  
  Use paths


Alright, we have a path. Now, we need to know how to use it.

  
    
  
  For example:


Let's take a value from nested structure
iex> first_username = path :usernames / 0
iex> Pathex.view(%{usernames: ["SubZero", "Scorpion"]}, first_username)
{:ok, "SubZero"}
Or let's change the value
iex> Pathex.set(%{usernames: ["SubZero", "Scorpion"]}, first_username, "Shao Khan")
{:ok, %{usernames: ["Shao Khan", "Scorpion"]}}
As you can see, every macro with the verbal name in Pathex module is a macro for using the path.
There are a lot of other ways to use paths.
In all such macros, the first argument is an input structure and the second argument is a path-closure.
Note:
Path-closure which are created inside path-using macros are optimised to have
only one operation generated (instead of default three).
You can read more about path-closures and operations here

As usual
	Macros without ! always return {:ok, result} or :error
	Macros with ! return result or raise Pathex.Error


  
    
  
  Compose paths


Path-closures can be composed together to create new path-closure,
every path composition macro is a binary operator. Some compositions
are optimized to generate one closure even if multiple closures
are used.
You can concat paths with Pathex.~>/2 composition macro

  
    
  
  For example:


iex> p1 = path :x
iex> p2 = path :y
iex> composed_path = p1 ~> p2
iex> 1 = Pathex.view!(%{x: [y: 1]}, composed_path)
You can think about path composition just like it's path concatenation in shell
For example:
iex> first_user = path :users / 0          # users/0
iex> name = path :name                     # name
iex> first_user_name = first_user ~> name  # users/0/name


  
    
  
  Prebuilt paths


Pathex provides some prebuilt paths for non-standard data manipulation. You
can find them in Pathex.Lenses module. You can read more about them in
lenses guide
For example Pathex.Lenses.star/0 lens works just like * in filesystem path.

  
    
  
  Performance tips


General rule: the more data you provide to Pathex at compile-time the better
For example:
path(1 / :x / :y, :json)
# works faster than
path(1 / :x / :y)
Because :json mod optimizes the closure to one big case,
while the default :naive mod generates nested cases.
And, in this example:
path(1 / :x / :y)
# works faster than
x = :x; path(1 / x / :y)
Because constants provide more information about available types.
This means that for :x Pathex knows that this can be a Keyword or Map,
while x means that this is a variable which could contain a List/Tuple index or a Map/Keyword value
And, in this example:
path(1 / x / :y)
# works faster than
path(1) ~> path(x) ~> path(:y)
Because path concatenation actually creates a path that calls all operands internally
which increases the call stack and makes concatenated paths handle errors manually.



  

    
Cheatsheet
    

This documentation page is intended to provide some common use cases of Pathex.

  
    
  
  Operations


This section describes actions which can be used with path-closures.
Note that for 1-depth operations Pathex is slightly slower than Map or Keyword
	Map	Pathex
	Map.fetch/2	Pathex.view/2
	Map.get/3	Pathex.get/3
	Map.update!/3	Pathex.over/3
	Map.update/4	Pathex.force_over/4
	Map.put/3	Pathex.force_set/3
	Map.replace/3	Pathex.set/3
	Map.delete/2	Pathex.delete/2
	Map.has_key?/2	Pathex.exists?/2


  
    
  
  Nested


This section describes how Pathex can be used instead of Access for accessing nested data structures.
Note that Pathex is 2 to 4 times faster than Access
	Access	Pathex
	get_in(s, [:x, :y, :z])	Pathex.get(s, path(:x / :y / :z))
	put_in(s, [:x, :y, :z], v)	Pathex.set(s, path(:x / :y / :z), v)

Note that Pathex works with structs (i.e. %User{}) as maps and doesn't need any special behaviour implemented in the module. Plus Pathex works with tuples and lists.
This means that this code
structure
|> get_in([:x, :y, :z])
|> Enum.at(10)
|> elem(1)
|> get_in([:z, :y, :x])
Can be rewritten to
Pathex.view(structure, path(:x / :y / :z / 10 / 1 / :z / :y / :x))

  
    
  
  Enumerables and lenses


	Enum	Pathex
	Enum.find/2	Pathex.Lenses.some/0
	Enum.map/2	Pathex.Lenses.all/0
	Enum.filter_map/3	Pathex.Lenses.star/0
	Enum.at/2	Pathex.path/2
	Enum.fetch/2	Pathex.path/2

More examples in Lenses tutorial

  
    
  
  Traverse leaves


This function traverses all leaves in the structure.
use Pathex; import Pathex.Combinator; import Pathex.Lenses

def leaves(iterlens \\ star()) do
  combine(fn recursive ->
    iterlens ~> (recursive ||| matching(_))
  end)
end

[2, 1, [:dot, 1234]] =
  %{
    x: 1,
    y: 2,
    meta: %{
      type: :dot,
      id: 1234
    }
  }
  |> Pathex.view!(leaves)
You can change the star() lens to whatever lens you prefer.
For example, for parsing HTML documents you can use star() ~> path(2) to
not traverse attributes. And if you want to find one leaf, you can use some()

  
    
  
  Walk structure


If you want to walk the whole structure, not only leaves, but the structure and
it's substructures too, you can use this function
use Pathex; import Pathex.Combinator; import Pathex.Lenses

# Like Macro.postwalk but for any tree-like structure
def postwalking(iterlens, predicate) do
  combine(fn recursive ->
    predicate
    ~> (
      alongside([
        iterlens ~> recursive,
        matching(_)
      ])
      ||| matching(_)
    )
    ||| (iterlens ~> recursive)
  end)
end

# Like Macro.prewalk but for any tree-like structure
def prewalking(iterlens, predicate) do
  combine(fn recursive ->
    predicate
    ~> (
      alongside([
        matching(_),
        iterlens ~> recursive
      ])
      ||| matching(_)
    )
    ||| (iterlens ~> recursive)
  end)
end

walking = postwalking(star(), matching(%{}))

# This code updates all maps and submaps in the structure
%{
  size: 3,
  x: 1,
  y: 2,
  meta: %{
    type: :dot,
    id: 1234,
    size: 3,
    empty_map_in_list: [[[%{size: 0}]]]
  }
} =
  %{
    x: 1,
    y: 2,
    meta: %{
      type: :dot,
      id: 1234,
      empty_map_in_list: [[[%{}]]]
    }
  }
  |> Pathex.over!(walking, & Map.put(&1, :size, map_size(&1)))



  

    
Lenses tutorial
    

This guide will show you how to create powerful lenses using Pathex.Lenses module. Here we will take a look at common tasks in nested data structure manipulation which can be solved using Pathex.Lenses.

  
    
  
  For all values in collection


What if we need to update all values in the collection matching specific pattern?
This simple task can be solved using Elixir's Enum module but it is quite tough
to be polymorphic and reusable for different patterns or types of collections
Let's say we have a list of users with roles and we want to add access to admin
page for all admins:
# The list looks like this
users = [
  %{fname: "John", sname: "Doe", role: "CEO",   access: ["admin_page", "users_page"]},
  %{fname: "Mike", sname: "Lee", role: "admin", access: ["users_page"]},
  %{fname: "Fred", sname: "Can", role: "admin", access: ["users_page"]},
  %{fname: "Dave", sname: "Lee", role: "user",  access: []}
]
With Enum this would look like
new_users =
  Enum.map(users, fn
    %{role: "admin", access: access} = user ->
      %{user | access: Enum.uniq(["admin_page" | access])}

    other ->
      other
  end)
But using Pathex.Lenses.star/0 and Pathex.Lenses.matching/1 this would look like
import Pathex
import Pathex.Lenses

# `l` in the end stands for `lens`
adminl = matching(%{role: "admin"})
accessl = path(:access)

# `star()` works like `*` in shell.
# so `star() ~> path(:file)` works almost like `*/file`

# Here `star() ~> adminl` translates to `select * where role == "admin"`
# and `star() ~> adminl ~> accessl` translates to `select access where role == "admin"`
new_users = Pathex.over!(users, star() ~> adminl ~> accessl, & Enum.uniq(["admin_page" | &1]))

  
    
  
  For any value in collection


What if we need to update first value matching specific pattern
(in our example it will be {:option, _}) and we need to
return {:ok, updated_collection} if the
first value was updated and :error if not
This task can be also done using Enum, but what if we can write the solution
which would be as simple as saying Update first value in collection, which matches the pattern?
With Enum this would look really terrible.
I couldn't come up with a polymorphic solution that would fit in less than 20 lines of code.
But with Pathex.Lenses.some/0 and Pathex.Lenses.matching/1 this would be as simple as
use Pathex; import Pathex.Lenses
def update_first_option(collection, update_func) do
  Pathex.over(collection, some() ~> matching({:option, _}), update_func)
end

  
    
  
  For any value in nested structure


Allright, we have a nested structure with various types inside and we need to find any value in any map
for which the special condition occurs and change it
Think of an HTML-like structure without attributes like
{"html", [
  {"head", [...]},
  {"body", [...]}
  ]}
And we need to update just one label with string which ends with "Please click subscribe button"
In Elixir we'd need to write a recursive function, which would untrivially update tuples and lists
Using Pathex.Combinator.combine/1, Pathex.Lenses.some/0 and Pathex.Lenses.filtering/1 it's very simple
use Pathex; import Pathex.Lenses; import Pathex.Lenses.Recur

path_to_subscribe =
  combine(fn recursive -> some() ~> (recursive() ||| matching()) end)
  ~> matching({"label", _}) # To find a label
  ~> path(1)               # To get to value of a label
  ~> filtering(& String.ends_with?(&1, "Please click subscribe button")

Pathex.set(document, path_to_subscribe, "Do not subscribe, hehe")

  
    
  
  Recurring


Sometimes you may want to create a recurring lens. We have a Pathex.Combinator.combine/1 which can help you. It is a fixed point combinator for paths, what may sound too smart, but this is a function which makes your lens be able to compose with itself.
Consider this representation of an HTML
{"html", [], [
  {"body", [], ...}
  {"html", [], ...}]}
And we want to update a node which has id=big-image. It would be nice to have a lens which would do something like
path(2) ~> some()
  ~> path(2) ~> some()
    ~> path(2) ~> some()`
Until it finds the tag we're looking for.
Meet the Pathex.Combinator.combine/1. Using this higher order function we can represent lens combining with itself.
The lens above can be specified with this code.
combine(fn recursive ->
  path(2) ~> some() ~> recursive
end)
And we also need to find a specific id, so let's combine this lens with matching(_)
path_to_id =
  combine(fn recursive ->
    path(2) ~> some ~> recursive
  end)
  ~> matching({_, [{"id", "big-image"}], _})
However, it has a drawback -- this recursive lens will never succeed, because we didn't specify the exit. Let's fix it
path_to_big_image =
  combine(fn recursive ->
    path(2) ~> some ~> (recursive ||| matching(_))
  end)
  ~> matching({_, [{"id", "big-image"}], _})

{"div", [{"id", "big-image"}], [
  {"img", [{"src", "/image.png"}], ""}]} = Pathex.view!(html, path_to_big_image)



  

    
Modifiers tutorial
    

Every pathex path created with Pathex.path/2 can have modifier specified as a second argument.
Modifier defines behaviour of the path in a way of structures it can match inside.
For example, path created with :map modifier can only match maps inside them.
Modifiers can be specified only in form of an atom, variables are not accepted.

  
    
  
  When? How? & Why?


You should use modifiers when you need to specify type of inner structures to match
or reduce amount of generated code by Pathex or improve performance of the path

  
    
  
  Usage


Currently only three modifiers are available:
	:json which matches lists and maps
	:naive which matches lists, tuples, keywords and maps
	:map which matches only maps

Default modifier for every path is :naive

Modifiers are specified as second argument in path/2 like
path :x / :y, :naive
path 0 / :x, :json

  
    
  
  Naive modifier


This modifier matches lists, tuples, keyword and maps
It generates matches for every structure like

  
    
  
  For example:


path(:x, :naive) generates something like
case input do
  %{x: value} ->
     ...
  [{a, _} | _] = k when is_atom(a) ->
     case Keyword.fetch(k, :x) do
       ...
     end
end

  
    
  
  Json modifier


This modifier specifies paths which macth lists (for integer keys only) and maps
Note:
This modifier treats variables as map keys, this means that
iex> x = 1
iex> p = path x, :json
iex> :error = Pathex.view([1, 2, 3], p)
iex> {:ok, :x} = Pathex.view(%{1 => :x}, p)

But passed integers are exanded into list matching
this makes it very efficient to view data from the structure

  
    
  
  For example:


path 1 / :x, :json generates closure with
case input do
  [_, %{x: value} | _] ->
    {:ok, value}

  %{1 => %{x: value}} ->
    {:ok, value}

  _ ->
    :error
end
Which extracts maximum efficiency from BEAM's pattern-matching

  
    
  
  Map modifier


This modifier matches only maps and therefore is the fastest modifier available

  
    
  
  For example:


path 1 / :x / "y", :map will generate closure with
case input do
  %{1 => %{x: %{"y" => value}}} ->
    {:ok, value}

  _ ->
    :error
end

  
    
  
  Annotations


Usually a key passed to path can match to one or more types. For example, key :x can be a key in Map like %{x: 1} and a key in Keyword like [x: 1]. To make key match only certain type it can be annotated to the exact type using this syntax path(:x :: :map). Available annotations are :map, :keyword, :list and :tuple.

  
    
  
  For example:


first_in_tuple = path(1 :: :tuple)
:error   = Pathex.view([0, 1, 2, 3], first_in_tuple)
{:ok, 0} = Pathex.view({0, 1, 2, 3}, first_in_tuple)



  

    
Path
    

Note:
This is a documentation about internal Pathex API and it is subject to change
You do not need to read this unless you are interested in Pathex internals, want to hack Pathex or want to create your own Pathex compatible lens

This page describes what a Pathex.t() is and how to create and use one.
Most of the time you might want to use Pathex.path/2 and Pathex.Lenses to create paths.
If you find their functionality limited and unapplicable for your use case, you can create you own Pathex-compatible closure. This doc describes how to do this.

  
    
  
  Path-closure


Path-closure is specified in Pathex.t(), and it's a closure of two arguments:
	Operation name. It is an atom, one of :view, :update, :force_update, :delete, :inpsect
	Operation arguments. It is a tuple which size depends on an operation

Currently every path-closure has 5 operations:
path_closure =
  fn
    # Operation which gets value from structure and return `function.(value)`
    :view,         {structure, hook_function} -> ...

    # Operation which returns new structure with updated value
    :update,       {structure, hook_function} -> ...

    # Operation which returns new structure with updated value, or default set
    :force_update, {structure, hook_function, default} -> ...

    # Operation which returns new structure with deleted value.
    :delete,       {structure, delete_function} -> ...

    # Inspects the path. Returns Elixir's AST. This is used only for error-logging and debugging
    :inspect,      _ -> ...
  end
Here structure is the structure which is viewed or updated by this path and hook_function or delete_function is a hook function described in the next section. If the value in the structure defined by the path is not present, path-closure must return :error. If it's present, path-closure must return whatever is returned by hook_function(value)

  
    
  
  Hook function


You can see that for every operation except inspect accepts some function as a second argument. This function is called a hook function and it must be called on the value from the structure defined by the path (if the value is present). Hook function is required to 

  
    
  
  Return types


Here function returns {:ok, result} | :error
And delete_function returns {:ok, result} | :error | :delete_me
	{:ok, result} returns updated value for update/force_update/delete operations and value to be returned for view
	:error in case function call has not succeeded
	:delete_me is returned by function only for :delete operation clause. It means that the value upon which the hook function was called must be deleted. For all other clauses, this must be treated as an invalid hook function and error must be raised


  
    
  
  Qualities


Special requirements are described here
	Path-closure must not raise or throw if it's called with correct operation and argument tuple

	Path-closure must be idempotent. This means that path-closure must return the same result for the same inputs every time it's called.

	Path-closure should not produce any side-effects. Thought it actually can produce side-effects, you shouldn't count on them.





  

    
Pathex 
    



      
This module contains main functions and macros used to create, use and manipulate paths.

  
    
  
  Usage


To use Pathex just insert to your context. You can import Pathex in module body or even in function body.
require Pathex
import Pathex, only: [path: 1, path: 2, "~>": 2, ...]
Or you can just use Pathex.
defmodule MyModule do

  # `default_mod` option is optional
  # when no mod is specified, `:naive` is selected
  use Pathex, default_mod: :json

  ...
end
This will import all operatiors and path macro

  
    
  
  Available macros


Any macro here belongs to one of three categories:
	Macro which creates path closure (only path/2)
	Macro which uses path closure to manipulate the value (like over/3, set/3, view/2, ...)
	Macro which creates some path composition (like alongside/1, ~>/2, |||/2, ...)


      


      
        Summary


  
    Types
  


    
      
        force_update_args(input, output)

      


    


    
      
        inner_func(output)

      


        Function which is passed to path-closure as second element in args tuple



    


    
      
        inspect_args()

      


    


    
      
        mod()

      


        More about modifiers



    


    
      
        pathex_compatible_structure()

      


        This depends on the modifier



    


    
      
        result(inner)

      


        Value returned by non-bang path call



    


    
      
        t()

      


        Also known as path-closure



    


    
      
        t(input, output)

      


        Also known as path-closure



    


    
      
        update_args(input, output)

      


    





  
    Functions
  


    
      
        
  
    
    Pathex.Accessibility - Pathex v2.5.2
    
    

    



  
  

    
Pathex.Accessibility 
    



      
Helpers to create paths from Access.t() and lists
Note:
This module is separate because functions presented in this module
are suboptimal. You should try use Pathex.path/2 first, and only
if its not applicable to you use-case, you should use functions
from this module


      


      
        Summary


  
    Types
  


    
      
        access()

      


    





  
    Functions
  


    
      
        from_access(access)

      


        Creates path-closure from Access.t().



    


    
      
        from_list(list, mod \\ :naive)

      


        Creates path from list of items. The list of items should not be known
at runtime, therefore some optimizations of paths are not possible. Use this
function only when Pathex.path/2 is not applicable



    


    
      
        from_record(module, macro, key)

      


        Creates a lens from record macro.
All arguments of this macro must be compile time atoms



    


    
      
        from_struct(module, key)

      


        Creates a lens for specific key in the structure. It behaves the same way as a lens created
with Pathex.path/2 but it works only with specified structure and force_set creates
a structure with default values.



    


    
      
        to_access(path_closure)

      


        Converts path-closure to Access.t()



    





      


      
        Types

        


  
    
      
      Link to this type
    
    access()


      
       
       View Source
     


  


  

      

          @type access() :: [term(), ...]


      



  


        

      

      
        Functions

        


  
    
      
      Link to this function
    
    from_access(access)


      
       
       View Source
     


  


  

      

          @spec from_access(access()) :: Pathex.t()


      


Creates path-closure from Access.t().
Note:
Paths created using this function do not support force operations yet


  
    
  
  Example


iex> import Pathex
iex> p = from_access [:x, :y]
iex> 10 = view!(%{x: [y: 10]}, p)

  



    

  
    
      
      Link to this function
    
    from_list(list, mod \\ :naive)


      
       
       View Source
     


  


  

      

          @spec from_list([any()], Pathex.mod()) :: Pathex.t()


      


Creates path from list of items. The list of items should not be known
at runtime, therefore some optimizations of paths are not possible. Use this
function only when Pathex.path/2 is not applicable

  
    
  
  Example


iex> import Pathex
iex> p = from_list [:x, 1, :y]
iex> 10 = view!(%{x: [1, [y: 10]]}, p)

  



  
    
      
      Link to this macro
    
    from_record(module, macro, key)


      
       
       View Source
     


      (macro)

  


  

Creates a lens from record macro.
All arguments of this macro must be compile time atoms

  
    
  
  Example


iex> require Pathex
iex> defmodule User do
...>   import Record; defrecord(:user, name: "", age: nil)
...>
...>   def new(name) do
...>     name_lens = from_record(User, :user, :name)
...>     Pathex.force_set!({}, name_lens, name)
...>   end
...> end
iex> {:user, "Joe", nil} = User.new("Joe")

  



  
    
      
      Link to this function
    
    from_struct(module, key)


      
       
       View Source
     


  


  

      

          @spec from_struct(module(), atom()) ::
  Pathex.t(
    struct(),
    any()
  )


      


Creates a lens for specific key in the structure. It behaves the same way as a lens created
with Pathex.path/2 but it works only with specified structure and force_set creates
a structure with default values.

  
    
  
  Example


iex> require Pathex
iex> defmodule User do
...>   defstruct name: "", age: nil
...> end
iex> name_lens = from_struct(User, :name)
iex> %{name: "Joe"} = Pathex.force_set!(%{}, name_lens, "Joe")

  



  
    
      
      Link to this function
    
    to_access(path_closure)


      
       
       View Source
     


  


  

      

          @spec to_access(Pathex.t()) :: [Access.access_fun(any(), any())]


      


Converts path-closure to Access.t()

  
    
  
  Example


iex> import Pathex
iex> access = to_access path(:x / 0, :map)
iex> 1 = get_in(%{x: %{0 => 1}}, access)

  


        

      



  

  
    
    Pathex.Combinator - Pathex v2.5.2
    
    

    



  
  

    
Pathex.Combinator 
    



      
Combinator for lenses
Read Pathex.Combinator.combine/1 documentation

      


      
        Summary


  
    Functions
  


    
      
        combine(path_func, max_depth \\ :infinity)

      


        This function creates a recursive path from path defined in path_func
Consider this example



    





      


      
        Functions

        


    

  
    
      
      Link to this function
    
    combine(path_func, max_depth \\ :infinity)


      
       
       View Source
     


  


  

      

          @spec combine((Pathex.t() -> Pathex.t()), pos_integer() | :infinity) :: Pathex.t()


      


This function creates a recursive path from path defined in path_func
Consider this example
iex> import Pathex; import Pathex.Lenses
iex> recursive_xpath = combine(fn recursive_xpath ->
iex>   path(:x)            # Takes by :x key
iex>   ~> recursive_xpath  # If taken, calls itself
iex>   ||| matching(_)     # Otherwise returns current structure
iex> end)
iex>
iex> Pathex.view!(%{x: %{x: %{x: %{x: 1}}}}, recursive_xpath)
1
iex> Pathex.set!(%{x: %{x: %{x: %{x: 1}}}}, recursive_xpath, 2)
%{x: %{x: %{x: %{x: 2}}}}
The second argument of this function specifies the maximum depth. It's infinity be default,
but you can specify this as any positive integer. It is useful when you're developing lens
and you're not sure whether the lens will or won't loop.
For example
# Combinator lens with limit
limited = combine(fn rec -> path(:x) ~> rec end, 100_000)
:error = Pathex.force_set(%{x: 1}, limited, 123)

# And this is without limit
unlimited = combine(fn rec -> path(:x) ~> rec end)
Pathex.force_set(%{x: 1}, unlimited, 123) # Infinite loop

  


        

      



  

  
    
    Pathex.Debug - Pathex v2.5.2
    
    

    



  
  

    
Pathex.Debug 
    



      
Special lens which logs all operation
and performs the same way matching(_) does

      


      
        Summary


  
    Functions
  


    
      
        debug(prefix \\ "")

      


    





      


      
        Functions

        


    

  
    
      
      Link to this function
    
    debug(prefix \\ "")


      
       
       View Source
     


  


  

      

          @spec debug(String.t()) :: Pathex.t()

Pathex.Lenses 
    



      
Module with collection of prebuilt paths

      


      
        Summary


  
    Functions
  


    
      
        all()

      


        Path function which works with all possible keys it can find
It takes all keys and than applies inner function (or concated path)
If any application fails, this lens returns :error



    


    
      
        any()

      


        Path function which works with any possible key it can find
It takes any key and than applies inner function (or concated path)



    


    
      
        filtering(predicate)

      


        This macro creates path-closure successes only when predicate returns truthy value.
predicate is a function which takes a structures upon which the path is called and
returns a boolean (or any other type which will be treated as boolean).



    


    
      
        matching(pattern)

      


        This macro creates path-closure which successes only when input matches the pattern.
The pattern can be just a pattern or a pattern with when. You can write this patterns
just like you'd write them in case



    


    
      
        some()

      


        Path function which applies inner function (or concated path-closure)
to the first value it can apply it to



    


    
      
        star()

      


        Path function which applies inner function (or concated path-closure)
to every value it can apply it to



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    all()


      
       
       View Source
     


  


  

      

          @spec all() :: Pathex.t()


      


Path function which works with all possible keys it can find
It takes all keys and than applies inner function (or concated path)
If any application fails, this lens returns :error

  
    
  
  Example


iex> require Pathex; import Pathex
iex> alll = Pathex.Lenses.all()
iex> Pathex.over!([%{x: 0}, [x: 1]], alll ~> path(:x), fn x -> x + 1 end)
[%{x: 1}, [x: 2]]
iex> Pathex.view!(%{x: 1, y: 2, z: 3}, alll) |> Enum.sort()
[1, 2, 3]
iex> Pathex.set([x: 1, y: 0], alll, 2)
{:ok, [x: 2, y: 2]}

  



  
    
      
      Link to this function
    
    any()


      
       
       View Source
     


  


  

      

          @spec any() :: Pathex.t()


      


Path function which works with any possible key it can find
It takes any key and than applies inner function (or concated path)

  
    
  
  Example


iex> require Pathex
iex> anyl = Pathex.Lenses.any()
iex> {:ok, 1} = Pathex.view %{x: 1}, anyl
iex> {:ok, [9]} = Pathex.set  [8], anyl, 9
iex> {:ok, [x: 1, y: 2]} = Pathex.force_set [x: 0, y: 2], anyl, 1
Note that force setting value to empty map has undefined behaviour
and therefore returns an error:
iex> require Pathex
iex> anyl = Pathex.Lenses.any()
iex> :error = Pathex.force_set(%{}, anyl, :well)
And note that this lens has keywords at head of list at a higher priority
than non-keyword heads:
iex> require Pathex
iex> anyl = Pathex.Lenses.any()
iex> {:ok, [{:x, 1}, 2]} = Pathex.set([{:x, 0}, 2], anyl, 1)
iex> {:ok, [1, {:x, 2}]} = Pathex.set([0, {:x, 2}], anyl, 1)
iex> {:ok, [1, 2]} = Pathex.set([{"some_tuple", "here"}, 2], anyl, 1)

  



  
    
      
      Link to this function
    
    filtering(predicate)


      
       
       View Source
     


  


  

      

          @spec filtering((any() -> boolean())) :: Pathex.t(any(), any())


      


This macro creates path-closure successes only when predicate returns truthy value.
predicate is a function which takes a structures upon which the path is called and
returns a boolean (or any other type which will be treated as boolean).
This function is useful when composed with star/0 and some/0
Note:
In terms of functional programming, such conditional lenses are called prisms


  
    
  
  Example


iex> import Pathex.Lenses; import Pathex
iex> adminl = filtering(& &1.role == :admin)
iex> {:ok, %{name: "Name", role: :admin}} = Pathex.view(%{name: "Name", role: :admin}, adminl)
iex> :error = Pathex.view(%{role: :user}, adminl)

iex> import Pathex.Lenses; import Pathex
iex> dots2d = [{1, 1}, {1, 5}, {3, 0}, {4, 3}]
iex> higher_than_2 = filtering(fn {_x, y} -> y > 2 end)
iex> {:ok, [{1, 5}, {4, 3}]} = Pathex.view(dots2d, star() ~> higher_than_2)

  



  
    
      
      Link to this macro
    
    matching(pattern)


      
       
       View Source
     


      (macro)

  


  

This macro creates path-closure which successes only when input matches the pattern.
The pattern can be just a pattern or a pattern with when. You can write this patterns
just like you'd write them in case
This function is useful when composed with star/0 and some/0
Note:
In terms of functional programming, such conditional lenses are called prisms


  
    
  
  Example


iex> import Pathex.Lenses; import Pathex
iex> adminl = matching(%{role: :admin})
iex> {:ok, %{name: "Name", role: :admin}} = Pathex.view(%{name: "Name", role: :admin}, adminl)
iex> :error = Pathex.view(%{}, adminl)

iex> import Pathex.Lenses; import Pathex
iex> dots2d = [{1, 1}, {1, 5}, {3, 0}, {4, 3}]
iex> higher_than_2 = matching({_x, y} when y > 2)
iex> {:ok, [{1, 5}, {4, 3}]} = Pathex.view(dots2d, star() ~> higher_than_2)

  



  
    
      
      Link to this function
    
    some()


      
       
       View Source
     


  


  

      

          @spec some() :: Pathex.t()


      


Path function which applies inner function (or concated path-closure)
to the first value it can apply it to

  
    
  
  Example


iex> require Pathex; import Pathex
iex> somel = Pathex.Lenses.some()
iex> Pathex.view!([x: [11], y: [22], z: 33], somel ~> path(0))
11
iex> Pathex.set!([x: %{y: 0}, z: %{y: 0}], somel ~> path(:y, :map), 1)
[x: %{y: 1}, z: %{y: 0}]
iex> Pathex.view([x: 1, y: 2, z: 3], somel)
{:ok, 1}
Note:
Force update fails for empty structures

Think of this function as star() ~> any() but optimized to work with only first element

  



  
    
      
      Link to this function
    
    star()


      
       
       View Source
     


  


  

      

          @spec star() :: Pathex.t()


      


Path function which applies inner function (or concated path-closure)
to every value it can apply it to

  
    
  
  Example


iex> require Pathex; import Pathex
iex> starl = Pathex.Lenses.star()
iex> Pathex.view!(%{x: [1], y: [2], z: 3}, starl ~> path(0)) |> Enum.sort()
[1, 2]
iex> Pathex.set!(%{x: %{y: 0}, z: [3]}, starl ~> path(:y, :map), 1)
%{x: %{y: 1}, z: [3]}
iex> Pathex.view([x: 1, y: 2, z: 3], starl)
{:ok, [1, 2, 3]}
Note:
It returns :error when no data was found or changed

Think of this function as filter_map. It is particularly useful for filtering
and selecting needed values with custom functions or matching/1 macro

  
    
  
  Example


iex> require Pathex; import Pathex; require Pathex.Lenses
iex> starl = Pathex.Lenses.star()
iex> structure = [{1, 4}, {2, 8}, {3, 6}, {4, 10}]
iex> #
iex> # For example we want to select all tuples with first element greater than 2
iex> #
iex> greater_than_2 = Pathex.Lenses.matching({x, _} when x > 2)
iex> Pathex.view(structure, starl ~> greater_than_2)
{:ok, [{3, 6}, {4, 10}]}

  


        

      



  

  
    
    Pathex.Short - Pathex v2.5.2
    
    

    



  
  

    
Pathex.Short 
    



      
This module provides short definitions of pathex paths.
For example, when using Pathex.Short, you can transform this
path :x / :y / 1 / 2
To just this
:x / :y / 1 / 2

      


      
        Summary


  
    Functions
  


    
      
        l / r

      


        This macro redefines / operator, so use this macro with caution
or only in bounded context (for example, you can use Pathex.Short only
inside functions)



    





      


      
        Functions

        


  
    
      
      Link to this macro
    
    l / r


      
       
       View Source
     


      (macro)

  


  

This macro redefines / operator, so use this macro with caution
or only in bounded context (for example, you can use Pathex.Short only
inside functions)

  
    
  
  Example


iex> use Pathex.Short
iex> path = :x / :y
iex> 1 = Pathex.view!(%{x: %{y: 1}}, path)

Pathex.Error exception
    



      
Simple exception for bang! functions (like Pathex.view!/2) errors.
Note:
Some new field may be added in the future


      


      
        Summary


  
    Types
  


    
      
        t()

      


    





      


      
        Types

        


  
    
      
      Link to this type
    
    t()


      
       
       View Source
     


  


  

      

          @type t() :: %Pathex.Error{
  __exception__: true,
  message: String.t(),
  op: atom(),
  path: Pathex.t(),
  structure: Pathex.pathex_compatible_structure()
}

